フィボナッチ数列をめぐるＪのコード

西川　利男

　別項で発表したデイジー、パイナップル、ひまわり、カリフラワーなどの葉序にはフィボナッチ数列がそのカギをにぎっている[1][2]。
　Ｊでフィボナッチ数列を求めるには、次の一行のコードを打ち込むだけでよい。直ちに、どこまででも数列は得られる。

```
Fibo =:  ], (,@(+/@(_2&{.)))

   Fibo ^:(15) 1 1
1 1 2 3 5 8 13 21 34 55 89 144 233 377 610 987 1597
```

このコーディングの意味も tacit を使っているが、それほど難しくはない。
「引数の後ろから２つの項をとり、その和をもとの引数の最後に追加する。」
これをパワー ^: により繰り返せば、数列はつぎつぎにと求められる。

また、各項の増加の割合、つまり隣り合った２つの項の比は１つの値に近づき、ふつう τ 値として数学のいろいろなところに顔を出す。

```
   FF =: Fibo ^:(15) 1 1
   Tau =: 2 ((%~)/)¥ FF

   Tau
1 2 1.5 1.66667 1.6 1.625 1.61538 1.61905 1.61765 1.61818 1.61798 1.61806
1.61803 1.61804 1.61803 1.61803
```

　Ｊの同じコーディングでも引数をつぎのように代えると、リュカの数列が得られる。

```
NB. Lucas Number
NB. Lucas =: Fibo^:(15) 2 1

   Fibo^:(15) 2 1
2 1 3 4 7 11 18 29 47 76 123 199 322 521 843 1364 2207
```

[1] J.H.コンウェイ、R.H.ガイ、根上生也訳「数の本」p.127-133
　　　　シュプリンガー・フェアラーク東京(2002).
[2] H.S.M.コクセター、銀林浩訳「幾何学入門」p.313-319 筑摩書房(2009).

ところが、たまたま「フィボナッチ」としてJ-Phrasesで検索してみたら、次のような記述を見つけた[3]。

[3] J Phrases
　　9　Mathematics: C: Polynomials and Rational Functions

Polynomial p. と Taylor t.を使ったものだが、なかなか手ごわいコーディングである。夏の合宿のおたのしみとして、知恵をしぼっていただきたい。

If f and g are polynomials, then the function f % g is called a rational function. The conjunction R=: 2 : 'x.&p. % y.&p.' (or R=: [. & p. % (]. & p.)) produces a rational function defined by its coefficient arguments:

```
   R=: [. & p. % (]. & p.)
   c=: 1 4 6 4 1 [ d=: 1 2 1  x=: i.6
   c R d
1 4 6 4 1&p. % 1 2 1&p.
   c R d x
1 4 9 16 25 36
   d R c x
1 0.25 0.1111111 0.0625 0.04 0.02777778
```

The Taylor coefficients of rational functions may provide interesting results. For example:

```
   c R d t. i. 10
1 2 1 0 0 0 0 0 0 0
   d R c t. i. 10
1 _2 3 _4 5 _6 7 _8 9 _10
   0 1 R 1 _1 _1 t. i. 10
0 1 1 2 3 5 8 13 21 34    NB. Fibonacci numbers
```

また、J Dictionaryのkには次のような例がある。

```
    f=: 1 2 1&p.
    g=: 1 3 3 1&p.
    (f*g) t. i.8
 1 5 10 10 5 1 0 0
    (f%g) t. i.8
 1 _1 1 _1 1 _1 1 _1
```

つまり、フィボナッチ数列を求めるには、次のようにすればよい。

```
   F =: 0  1&p.
   G =: 1 _1 _1&p.
   (F%G) t. i.10
0 1 1 2 3 5 8 13 21 34
```

いろいろ本を漁った結果、Ｊのこのコーディングを理解するには、母関数について知ることが必要であることがわかった。

母関数 (Generating Function)とは、次のようなものである。

数列 $\{a_n\}\ (n = 0,\ 1, \cdots)$ を調べる際に、

これらを係数とする級数 $f(x) = a_0 + a_1 x + a_2 x^2 + \cdots$

を考えると見通しよくなることがある。その級数関数$f(x)$を数列$\{a_n\}\ (n = 0,\ 1, \cdots)$の母関数（あるいは生成関数）という。…… 岩波数学入門辞典 p.569

フィボナッチ数列０ １ １ ２ ３ ５ ８ … についてやってみる。
これに対する母関数は

$$F(x) = F_1 x + F_2 x^2 + F_3 x^3 + F_4 x^4 + F_5 x^5 + F_6 x^6 \cdots$$
$$= x + x^2 + 2x^3 + 3x^4 + 5x^5 + 8x^6 \cdots$$

となるが、これに対するもっと簡単な式を求めてみよう。
漸化式

$$F_n = F_{n-1} + F_{n-2}$$

を使って、次のように置き換える。

$$F(x) = F_1 x + F_2 x^2 + (F_2 + F_1)x^3 + (F_3 + F_2)x^4 + (F_4 + F_3)x^5 + (F_5 + F_4)x^6 \cdots$$

かっこ内の後ろ同士、前同士をまとめる。そして、整理すると

$$F(x) = F_1 x + F_2 x^2 + \{F_1 x^3 + F_2 x^4 + F_3 x^5 + F_4 x^6 + \cdots\}$$
$$+ (F_2 x^3 + F_3 x^4 + F_4 x^5 + F_5 x^6 + \cdots\}$$
$$= F_1 x + F_2 x^2 + x^2\{F_1 x + F_2 x^2 + \cdots\} + x\{F_2 x^2 + F_3 x^3 + \cdots\}$$

$F_1 = 1,\ F_2 = 1$として、前のかっこ部分は$F(x)$、後ろのかっこ部分は$F(x) - F_1 x$にそれぞれ等しいので、次の式になる。

$$F(x) = x + x^2 + x^2 F(x) + x(F(x) - x)$$
$$= x + x^2 F(x) + xF(x)$$

これを$F(x)$について解くと、最終的に次の式が得られる[4]。

$$F(x) = \frac{x}{1 - x - x^2}$$

[4] J.H.シルヴァーマン、鈴木治郎訳「はじめての数論」p.290-294,
ピアソン・エデュケーション(2003).

Ｊのコーディングはこの式に基づいている。この分数式にTaylor t.を利用するとべき級数の係数として、フィボナッチ数列が得られるという仕組みである。

```
NB. From t. in J-Dictionary ===========================
  f=: 1 2 1&p.
  f
+-----+-+--+
|1 2 1|&|p.|
+-----+-+--+
  g=: 1 3 3 1&p.
  g
+-------+-+--+
|1 3 3 1|&|p.|
+-------+-+--+
   f*g
+-+-+-+
|f|*|g|
+-+-+-+

   i.8
0 1 2 3 4 5 6 7
    f t. i.8
1 2 1 0 0 0 0 0
    g t. i.8
1 3 3 1 0 0 0 0
    (f*g) t. i.8
1 5 10 10 5 1 0 0
    (f%g) t. i.8
1 _1 1 _1 1 _1 1 _1
NB. Applied to Fibonacci =================================
   F =: 0  1&p.
   G =: 1 _1 _1&p.

   F
+---+-+--+
|0 1|&|p.|
+---+-+--+
   G
+-------+-+--+
|1 _1 _1|&|p.|
+-------+-+--+
   F t. i.10
0 1 0 0 0 0 0 0 0 0
   G t. i.10
1 _1 _1 0 0 0 0 0 0 0
   (F%G) t. i.10
0 1 1 2 3 5 8 13 21 34
```